

Formuły tablicowe

W wielu zagadnieniach do obliczeń można wykorzystać nie jedną (jak do tej pory), ale wiele wartości równocześnie. Zamiast tworzyć wiele identycznych formuł (ze zmieniającymi się tylko adresami komórek) możemy przy pomocy jednej **formuły tablicowej** wykonać skomplikowane obliczenia. Zestawy wartości do obliczeń w formułach tablicowych najczęściej przechowywane są w tablicach, które EXCEL przechowuje na dwa sposoby:

- w **komórkach** arkusza (np. A1:B10)
- w postaci **stałej tablicowej**, zawierającej wiele elementów, np. { 10;20;30;40\50;60;70;80}

Zalety wykorzystania formuł tablicowych:

- zużywają mniej pamięci operacyjnej komputera, a pliki są mniejsze
- w dowolną komórkę wyświetla się ta sama formuła
- bezpieczeństwo - nie można usunąć pojedynczej formuły
- mogą działać bardziej efektywnie
- mogą wyeliminować potrzebę stosowania formuł pośrednich
- umożliwiają wykonanie pewnych operacji, których bez użycia tablic nie da się wykonać

Wady:

- wydłuża czas przeliczania formuł w arkuszu na bardzo dużych tablicach
- arkusze stają się dla innych użytkowników trudniejsze do zrozumienia

Reguły używania formuł tablicowych

- lub zaznaczyć obszar, który będzie zawierał wyniki formuły,
- wpisać treść formuły tablicowej,
- wcisnąć kombinację klawiszy **CTRL + +SHIFT + ENTER**
- po poprawnym wykonaniu formuła powinna być otoczona parą nawiasów klamrowych { }
- nie można przenosić pojedynczych komórek z obszaru wyników formuły tablicowej – można całość
- z obszaru wyników formuły tablicowej nie można usuwać pojedynczych komórek – można całość
- wszystkie tablice używane w formułach tablicowych powinny mieć takie same wymiary

Kiedy użyć funkcji tablicowej?

Oczywiście można nigdy nie używać takich formuł, ale wtedy wypełniamy arkusz wieloma formułami ze skomplikowanym adresowaniem.

PRZYKŁAD – **sposób tradycyjny** – wyliczamy wartość brutto w kolumnie B, wklejając do kolejnych komórek wzór: $=A3+A3*\$B\1

			=A3+A3*\$B\$1
	A	B	C
1	vat	8%	
2	cena	brutto	
3	100	108	
4	120	129,6	
5	145	156,6	
6	200	216	

PRZYKŁAD – sposób tablicowy – zaznaczamy zakres komórek i wpisujemy wyrażenie: $=A3:A6+A6*B1$ a następnie naciskamy kombinację klawiszy CTRL + SHIFT + ENTER. We wszystkich zaznaczonych komórkach zostanie wyliczona odpowiednia wartość. W języku matematyki można powiedzieć, że wektor mnożymy przez stałą, czyli każdą współrzędną wektora (komórkę tablicy A3:A6) mnożymy przez komórkę B1.

B3		fx {=A3:A6+A3:A6*B1}		
	A	B	C	D
1	vat	8%		
2	cena	brutto		
3	100	108		
4	120	129,6		
5	145	156,6		
6	200	216		

Przykład powyższy pokazywał mnożenie wektora przez liczbę, w wyniku czego otrzymaliśmy wektor. Przykład kolejny dotyczy mnożenia dwóch wektorów – każdy element pierwszego wektora jest mnożony przez każdy element drugiego wektora, w wyniku czego powstanie tablica.

Formuła tablicowa dająca w wyniku tablicę – wiele wyników

PRZYKŁAD – **tabliczka mnożenia** – mnożymy przez siebie dwa wektory: pionowy A2:A11 i poziomy B1:K1.

- zaznaczamy obszar komórek B2 do K11
- wpisujemy do aktywnej komórki formułę =B1:K1*A2:A11
- naciskamy CTRL + ALT+ ENTER

Nie musimy używać adresów mieszanych, gdybyśmy zdecydowali się na tradycyjne kopiowanie i wklejanie formuły:

=B\$1*\$A2

[illegible]

PRZYKŁAD – zaokrąglanie - w pewnym obszarze komórek znajdują się liczby rzeczywiste. Za pomocą funkcji ZAOKR.DO.CAŁK i formuły tablicowej wszystkie wartości zostaną zaokrąglone w nowych komórkach

- zaznaczamy identyczny obszar komórek
- wpisujemy formułę: =ZAOKR.DO.CAŁK(A1:D4)
- naciskamy klawisze CTRL + SHIFT + ENTER

Otrzymujemy tablicę z wynikami. Tradycyjny sposób obliczania polegałby na wpisaniu formuły =ZAOKR.DO.CAŁK(A1) i skopiowaniu jej do pozostałych komórek.

	A6		fx {=ZAOKR.DO.CALK(A1:D4)}		
	A	B	C	D	E
1	19,72835	860,7017	1,893325	103,6666	
2	115,7184	51,74416	149,9265	98,75382	
3	14,35004	208,3644	145,939	7422,819	
4	7,829993	162,5201	359,5365	3834,428	
5					
6	19	860	1	103	
7	115	51	149	98	
8	14	208	145	7422	
9	7	162	359	3834	
10					

Formuła tablicowa dająca jeden wynik

PRZYKŁAD – średnia ważona ocen – nauczyciel informatyki stosuje tzw. „wagi ocen”, tzn. każda ocena, w zależności od swojej ważności (sprawdzian pisemny, kartkówka, odpowiedź ustna, zadanie domowe itp.) posiada dodatkowe oznaczenie liczbowe. Zwyczajna średnia arytmetyczna z ocen 3, 4, 5 da w wyniku 4. Tak się dzieje, jeżeli domyślnie waga każdej oceny jest jednakowa (wynosi np. jeden). Jeżeli oceny będą miały różne wagi, nie będziemy liczyli średniej arytmetycznej lecz średnią ważoną: suma iloczynów (ocena * waga) podzielona przez sumę wag.

Jeżeli (dla uproszczenia zwykłej formuły) w komórkach kolumny D znajdują się iloczyny komórek z kolumn A i B, np. w D2: =A2*B2 (i podobnie w następnych), to tradycyjna formuła w komórce D6 mogła by mieć postać: =SUMA(D2:D4)/SUMA(B2:B4). Bez posługiwania się dodatkowymi iloczynami: {=SUMA(A2*B2;A3*B3;A4*B4)/SUMA(B2:B4)}

Za pomocą formuły tablicowej

- wpisujemy do komórki formułę: =SUMA(A2:A4*B2:B4)/SUMA(B2:B4)
- naciskamy CTRL + SHIFT + ENTER

Przykład stanie się bardziej wymowny, jeżeli znajdzie konieczność wyliczenia średniej nie trzech, ale np. trzydziestu ocen. W formule mnożona jest tablica jednowymiarowa A2:A4 przez identyczną tablicę w komórkach B2:B4. Tablice, na których wykonujemy obliczenia tablicowe muszą być identyczne.

	A	B	C	D
1	ocena	waga		iloczyn
2	5	2		10
3	4	1		4
4	3	0,5		1,5
5				
6				4,428571
7				śr. ważona

Stałe tablicowe

Wartości do policzenia mogą znajdować się także w wektorach liczbowych.

PRZYKŁAD - tabliczka mnożenia – mnożymy przez siebie dwie stałe tablicowe (dwa wektory). W pierwszej liczby ułożone są poziomo (rozdzielone znakiem „;”), w drugiej pionowo (rozdzielone znakiem „\”).

- zaznacz komórki od A1 do J10 (obszar o wielkości 10 na 10)
- kliknij w pasek formuły (edycji) lub naciśnij F2
- wpisz (lub wklej) formułę tablicową: {=1;2;3;4;5;6;7;8;9;10}*{1\2\3\4\5\6\7\8\9\10}
- na klawiaturze naciśnij CTRL + SHIFT + ENTER

We wszystkich zaznaczonych komórkach pojawi się formuła tablicowa: {=1;2;3;4;5;6;7;8;9;10}*{1\2\3\4\5\6\7\8\9\10} i odpowiadające jej odpowiednie wartości tabliczki mnożenia. Liczby ze stałej tablicowej pionowej mnożone są przez liczby ze stałej poziomej: każda przez każdą, dlatego w wyniku otrzymujemy tablicę 100 wartości.

A	B	C	D	E	F	G	H	I	J
1	1	2	3	4	5	6	7	8	9
2	2	4	6	8	10	12	14	16	18
3	3	6	9	12	15	18	21	24	27
4	4	8	12	16	20	24	28	32	36
5	5	10	15	20	25	30	35	40	45
6	6	12	18	24	30	36	42	48	54
7	7	14	21	28	35	42	49	56	63
8	8	16	24	32	40	48	56	64	72
9	9	18	27	36	45	54	63	72	81
10	10	20	30	40	50	60	70	80	90

Definiowanie nazwy stałej tablicowej

Jeżeli stała tablicowa będzie wykorzystywana wielokrotnie, to zamiast wpisywać za każdym razem całość, można ją zdefiniować za pomocą okna dialogowego DEFINIOWANIE NAZW (Excel 2003: Wstaw-Nazwa-Definiuj, Excel 2007: Formuły-Definiuj nazwę).

Jeżeli zdefiniujemy w ten sposób dwie nazwy pionowo: {=1;2;3;4;5;6;7;8;9;10} i poziomo: {=1\2\3\4\5\6\7\8\9\10}, będziemy mogli do komórek wpisać wyrażenie tablicowe: =pionowo*poziomo.

Po naciśnięciu kombinacji klawiszy CTRL + SHIFT + ENTER uzyskamy identyczny wynik i formułę tablicową: {=pionowo*poziomo}.

PRZYKŁAD - kwota brutto – ceny netto znajdujące się w kolumnie A należy powiększyć o stawkę VAT w kolumnie B. Tablicę komórek A2:A5 mnożymy przez tablicę komórek B2:B5 zamienioną na procent i powiększoną o 1.

- zaznaczamy komórki D2:D5
- wpisujemy formułę tablicową $=A2:A5*(1+B2:B5\%)$
- naciskamy klawisze CTRL + SHIFT + ENTER

Otrzymujemy tablicę jednowymiarową z wynikami. Tradycyjny sposób obliczania polegałby na wpisaniu formuły: $A2*(1+B2\%)$ i skopiowaniu jej do następnych komórek.

	A	B	C	D	E	F
1	cena	vat		brutto		
2	100	7		107		
3	123	7		131,61		
4	221	23		271,83		
5	130	5		136,5		

Definiowanie nazwy stałej tablicowej

Podobnie, jak poprzednio, gdy definiowaliśmy stałe tablicowe związane z konkretnymi wartościami, możemy zdefiniować nazwy związane z tablicą komórek. Wykorzystamy poprzedni przykład.

- zaznaczamy komórki A2 do A5
- wywołujemy okno dialogowe Definiuj nazwę
 - odpowiednie informacje wpisane są automatycznie lub wpisujemy nowe

	A	B	C	D	E
1	cena	vat		brutto	
2	100	7		107	
3	123	7		131,61	
4	221	23		271,83	
5	130	5		136,5	

Nazwa:

Zakres:

Komentarz:

$=\text{Arkusz2!\$A\$2:\$A\$5}$

OK Anuluj

W podobny sposób definiujemy wektor VAT, a formuła tablicowa przyjmie postać: $=\text{cena}*(1+\text{vat}\%)$

PRZYKŁAD - kwota brutto – podobny sposób obliczania z wykorzystaniem tylko jednej wartości stawki VAT. Tablicę komórek A2:A5 mnożymy przez 1,07

	A	B	C	D	E
1	cena	vat	7		
2	100			107	
3	123			131,61	
4	221			236,47	
5	130			139,1	

Coś bardziej skomplikowanego

W kolumnie oznaczonej CZĘŚCI użytkownik wpisuje numery pewnych części. Numery te mogą się powtarzać i mogą być w przedziale 1 do 1000. Naszym zadaniem jest policzyć ile różnych części wpisano. Jeżeli w kolumnie są następujące numery: 1,2,5,3,5,2,1, to wynik wynosi 4 (bo 4 różne numery: 1,2,3,5). Zadanie można wykonać tradycyjnie używając funkcji licz.jeżeli, jeżeli, suma itp. lub za pomocą licz.jeżeli i tabeli przestawnej albo też używając wyrażeń tablicowych.

	A	B	C	D	E	F	G	H	I
1									
2	CZĘŚCI								
3	1								
4	2								
5	5								
6									
7	3								
8	5								
9									
10	2								
11	1								
12									

Formuła tablicowa:

$=\text{SUMA}(\text{JEŻELI}(\text{LICZ.JEŻELI}(\text{A5:A1004}; \text{WIERSZ}(\text{A1:A1000}))>0;1;0))$

WIERSZ(A1:A1000) tworzy wektor, z kolejnymi liczbami naturalnymi od 1 do 1000.

LICZ.JEŻELI(A5:A1004; ...) porównuje numery części z liczbami naturalnymi i zlicza ilość wystąpień numerów części, np. część o numerze 5 występuje 2 razy

JEŻELI(... >0;1;0) jeżeli dany numer części występuje więcej niż 1 raz, to w nowym wektorze wpisuje 1, w przeciwnym wypadku wpisuje 0 (każdy numer części, który występuje na liście ma przypisaną jedynkę)

SUMA(...) sumujemy jedynki i otrzymujemy ilość wpisanych części

Na obrazku obok przedstawiono kolejne operacje, które wykonywane są na wektorach w pamięci komputera.

CZĘŚCI	WIERSZ	LICZ.JEŻELI	JEŻELI	SUMA
1	1	2	1	4
2	2	2	1	
5	3	1	1	
	4	0	0	
3	5	2	1	
5	6	0	0	
	7	0	0	
2	8	0	0	
1	9	0	0	
	10	0	0	

ZADANIA

Przeanalizuj poniższe przykłady samodzielnie i wykonaj na arkusz STUDENCI.XLS

KOLUMNY

W nowej kolumnie utwórz skrót składający się z trzech pierwszych liter nazwiska dużymi literami i trzech pierwszych imienia małymi literami (np. Libront Wacław – LIBwac)

=LITERY.WIELKIE(LEWY(D2:D374;3))&LITERY.MAŁE(LEWY(D2:D374;3))

W nowej kolumnie wypisz nazwy ulic bez numerów domów

=LEWY(E2:E374;SZUKAJ.TEKST(" ";E2:E374)-1)

Niektóre ulice mają niepoprawnie wpisane numery domów – bez znaku odstępu, dlatego poprzednia formuła powoduje błąd (np. wiersz 57). Skorzystaj z instrukcji warunkowej i funkcji CZY.BŁĄD i popraw formułę, aby dawała zawsze wyniki poprawne

**=JEŻELI(CZY.BŁĄD(SZUKAJ.TEKST(" ";E2:E374));E2:E374;
LEWY(E2:E374;SZUKAJ.TEKST(" ";E2:E374)-1))**

Wskaźnik OCENA*PUNKTY stanowi kryterium, według którego studenci będą przyjmowani na uczelnię. Wylicz ten wskaźnik każdemu studentowi, który został przyjęty na uczelnię

=JEŻELI(Q2:Q374="tak";N2:N374*T2:T374;"")

KOMÓRKI

W nowej komórce wylicz ilu studentów urodziło się w sierpniu

=SUMA(JEŻELI(MIESIĄC(G2:G374)=8;1;0))

W nowej komórce wylicz ilu studentów ukończyło LO

=SUMA(JEŻELI(I2:I374="LO";1;0))

Jaka jest średnia arytmetyczna wszystkich wskaźników OCENA*PUNKTY. Wylicz za pomocą jednej formuły w jednej komórce

=ŚREDNIA(JEŻELI(Q2:Q374="tak";N2:N374*T2:T374;""))

OBSZAR

Od kwoty dochodu studenta będą pobierane co miesiąc niewielkie składki. W kolejnych czterech kolumnach chcemy policzyć wielkość tej składki każdemu studentowi dla: 1%; 1,1%; 1,2% i 1,3%

=U2:U374*{0,01;0,011;0,012;0,013}

UWAGA – należy zaznaczyć od razu cztery kolumny